AIM-Spice Reference Manual

Contents

Introduction 1

Notation Used in the Manual 1

AC Analysis 2

DC Operating Point 3

DC Transfer Curve Analysis 4

Distortion Analysis (Not available yet) 5

Noise Analysis 7

Pole/Zero Analysis 8

Transfer Function Analysis 9

Transient Analysis 10

Options 11

Title Line 13

Comment line 14

A Heterostructure Field Effect Transistors (HFETs) 15

B Non-linear Dependent Sources 17

C Capacitors 19

D Diodes 21

E Linear Voltage-Controlled Voltage Sources 23

F Linear Current-Controlled Current Sources 24

G Linear Voltage-Controlled Current Sources 25

H Linear Current-Controlled Voltage Sources 26

I Independent Current Sources 27

J Junction Field-Effect Transistors (JFETs) 33

K Coupled Inductors (transformers) 35

L Inductors 36

M Metal Oxide Semiconductor Field Effect Transistors (MOSFETs) 37

N Heterojunction Bipolar Transistors (HBTs) 47

O Lossy Transmission Lines (LTRA) 50

Q Bipolar Junction Transistors (BJTs) 52

R Resistors 55

S Voltage Controlled Switch 57
T Transmission Lines (Lossless) 59
U Uniform Distributed RC Lines (URC) 60
V Independent Voltage Sources 61
W Current Controlled Switch 67
Z GaAs Metal Semiconductor Field Effect Transistors (MESFETs) 69
.SUBCKT statement 73
.ENDS statement 74
Call to subcircuits 75
.NODESET statement 76
.IC statement 77
Bugs Reported by Berkeley 78
References 74

Introduction

This reference manual contains a complete listing of all analyses, options, devices and control statements in AIM-Spice. Analyses and Options are specified by choosing menu commands, and devices and control statements are specified in the circuit description. The reference first lists all analyses and options, then devices and control statements.

Notation Used in the Manual

Item	Example	Description
name	M12	A name field is an alphanumeric string. It must begin with a letter and cannot contain any delimiters.
node	5000	A node field may be arbitrary character strings. The ground node must be named '0'. Node names are treated as character strings, thus '0' and '000' are different names.
scale suffix		$T=10^{12}$, $G=10^{9}$, $MEG=10^{6}$, $K=10^{3}$, $MIL=25.4\cdot 10^{-6}$, $M=10^{-3}$, $U=10^{-6}$, $N=10^{-9}$, $P=10^{-12}$, $F=10^{-15}$
units suffix	V	Any letter that is not a scale factor or any letters that follows a scale suffix
value	1KHz	Floating-point number with optional scale and/or units suffixes
(text)	(option)	Comment
<item>	<OFF>	Optional item
{item}	{model}	Required item

AC Analysis

AC analysis is used to calculate the frequency response of a circuit over a range of frequencies.



DEC stands for decade variation, OCT stands for octave variation, and LIN stands for linear variation. The specification of number of points changes with the selection of DEC, OCT or LIN. If DEC is specified, the number of points are pr. decade. If OCT is specified, the number of points are pr. octave, and if LIN is specified the number of points are the total number across the whole frequency range. The frequency range is specified with the Start Frequency and End Frequency parameters. Note that in order for this analysis to be meaningful, at least one independent source must have been specified with an ac value.

If the circuit has only one ac input, it is convenient to set that input to unity and zero phase. Then the output variable will be the transfer function of the output variable with respect to the input.

DC Operating Point

This analysis calculates the dc operating point of a circuit. It has no parameters.

DC Transfer Curve Analysis

In a DC Transfer Curve analysis, one or two source(s) (voltage or current source) are swept over a user defined interval. The dc operating point of the circuit is calculated for every value of the source(s).



Source Name is the name of an independent voltage or current source, Start Value, End Value and Increment Value are the starting, final and increment values respectively. The example above causes the voltage source vds to be swept from 0 Volts to 5 Volts in increments of 0.1 Volts. A second source may optionally be specified with associated sweep parameters. In this case, the first source is swept over its range for each value of the second source. This option can be useful for obtaining semiconductor device output characteristics.

Distortion Analysis (Not available yet)

The distortion analysis part of AIM-Spice calculates steady-state harmonic and intermodulation products for small signal input magnitudes.



The dialog box for this analysis is almost identical to the AC analysis dialog box except for the last parameter F2 over F1. This is an optional parameter and if not specified, the distortion analysis performs a harmonic analysis - i.e., it analyses distortion in the circuit using only a single input frequency F1, which is swept as specified by the frequency parameters exactly as in the AC analysis. Inputs at this frequency may be present at more than one input source, and their magnitudes and phases are specified by the arguments of the DISTOF1 keyword on the instance lines for the input sources (see the description for independent sources). The analysis produce information about the AC values of all node voltages and branch currents at the harmonic frequencies 2F1 and 3F1, vs. the input frequency F1 as it is swept. (A value of 1 (as a complex distortion output) signifies $\cos(2\pi(2F1)t)$ at 2F1 and $\cos(2\pi(3F1)t)$ at 3F1, using the convention that 1 at the input fundamental frequency is equivalent to $\cos(2\pi F1t)$.) The distortion component desired (2F1 or 3F1) can be selected using the postprocessor, and then printed. (Normally, one is interested primarily in the magnitude of the harmonic components, so the magnitude of the AC distortion value is looked at). It should be noted that these are the AC values of the actual harmonic components, and are not equal to HD2 and HD3. To obtain HD2 and HD3, one must divide by the corresponding AC values at F1, obtained from an AC analysis.

If the optional F2 over F1 parameter is specified, it should be a real number between (and not equal to) 0.0 and 1.0. In this case a spectral analysis is performed. It considers the circuit with sinusoidal inputs at two different frequencies F1 and F2. F1 is swept according to the frequency parameters in the dialog box. F2 is kept fixed at a single frequency as F1 sweeps - the value at which it is kept fixed is equal to "F2 over F1" times the Start frequency. Each independent source in the circuit may potentially have two (superimposed) sinusoidal inputs for distortion, at the frequencies F1 and F2. The magnitude and phase of the F1 component are specified by the arguments of the DISTOF1 keyword in the source's element line (see the description of independent sources). The magnitude and phase of the F2 component are specified by the arguments of the DISTOF2 keyword. The analysis produces plots of all node voltages/branch currents at the intermodulation product frequencies F1+F2, F1-F2, and 2F1 - F2, vs. the swept frequency F1. The IM product of interest may be selected and displayed/printed in the postprocessor.. It is to be noted as in the harmonic analysis case, the results are the actual AC voltages and currents at the intermodulation frequencies, and need to be normalised with respect to .AC values to obtain the IM parameters.

If the DISTOF1 or DISTOF2 keywords are missing from the description of an independent source, then that source is assumed to have no input at the corresponding frequency. The default values of the magnitude and phase are 1.0 and 0.0 respectively. The phase should be specified in degrees.

It should be carefully noted that the number specified for the F2 over F1 parameter should ideally be an irrational number, and since this is not possible in practice, efforts should be made to keep the denominator in its fractional representation as large as possible, certainly above 3, for accurate results (i.e. the parameter is represented as a fraction A/B where A and B are integers with no common factors, B should be as large as possible. Note that A<B because the parameter is constrained to be < 1). To illustrate why, consider the cases where the parameter is 49/100 and 1/2. In a spectral analysis, the outputs produced are at F1+F2, F1-F2 and 2F1 -F2. In the latter case, F1-F2=F2, so the results at the F1-F2 component is erroneous because there is the strong fundamental F2 component at the same frequency. Also, F1+F2=2F1-F2 in the latter case, and each result is erroneous individually. This problem is not there in the case where the parameter "F2 over F1" is 49/100. In this case, there are two very closely spaced frequency components at F2 and F1-F2. One of the advantages of the Volterra series technique is that it computes distortions at mix frequencies expressed symbolically, therefore one is able to obtain the strengths of distortion components accurately even if the separation between them is very small, as opposed to the transient analysis for example. The disadvantage is of course that if two of the mix frequencies coincide, the results are not merged together and presented. Currently, the interested user should keep track of this mix frequencies himself or herself and add the distortions at coinciding mix frequencies together should it be necessary.

Noise Analysis

The noise analysis portion of AIM-Spice computes device-generated noise for the given circuit.



The "Output Noise Variable" parameter has the form V(OUTPUT<,REF>) where OUTPUT is the node at which the total output noise is desired. If REF is specified, the noise voltage V(OUTPUT) - V(REF) is calculated. By default REF is assumed to be ground. The "Input Source" parameter is the name of an independent source to which input noise is referred. The next three parameters are the same as for AC analysis. The last parameter is an optional integer; if specified, the noise contribution of each noise generator is produced every "Points per Summary" frequency points.

This analysis produces two plots. One for the Noise Spectral Density curves and one for the total Integrated Noise over the specified frequency range. All noise voltages/currents are in squared units (V^2/Hz and A^2/Hz for spectral density, V^2 and A^2 for integrated noise).

Pole/Zero Analysis

The pole-zero analysis computes poles and/or zeros in the small signal ac transfer function.



You can specify that AIM-Spice should locate only poles or zeros. This feature is provided mainly because if there is a non convergence in finding the poles or zeros, then, at least the other can be found.

Transfer Function Analysis

When you select this analysis AIM-Spice computes the DC small signal value of the transfer function, input resistance, and output resistance.



With the example shown in the dialog box above, AIM-Spice will compute the ratio v(8) to vin, the small-signal input resistance at vin, and the small-signal output resistance measured across nodes 8 and 0.

Transient Analysis

The transient analysis in AIM-Spice computes the time domain response of a circuit.



The parameter Stepsize is the suggested computing increment, Final Time is the last time point computed. The transient analysis always starts at time zero. If you are not interested in the results until a time t_1 greater than zero, you specify t_1 for the parameter Display Start Time. The parameter Maximum Stepsize is useful when you want to limit the internal stepsize used by AIM-Spice. The Use Initial Conditions (UIC) option, when specified, indicates that the user does not want AIM-Spice to solve for the quiescent operating point before beginning the transient analysis. The values specified using IC= on the various elements are used as the initial transient condition. If the .IC control line has been specified, then the node voltages on the .IC line are used to compute the initial conditions for the devices. Look at the description on the .IC control line for its interpretation when UIC is not specified.

Options

A set of options that controls different aspects of a simulation is available through the Options menu. The options are divided between four logical groups as shown below.

- General
- Analysis specific
- Device specific
- Numeric specific

Each option group have their own dialog box. The options are listed in the table below.

Option	Description	Default
GMIN	Minimum allowed conductance (1)	1.0E-12
RELTOL	Relative error tolerance (1)	0.001
ABSTOL	Absolute current error tolerance (1)	1nA
VNTOL	Absolute voltage error tolerance (1)	1μV
CHGTOL	Charge tolerance (1)	1.0E-14
TNOM	Nominal temperature. The value can be overridden by a temperature specification on any temperature dependent device model. (1)	27
TEMP	Operating temperature of the circuit. The value can be overridden by a temperature specification on any temperature dependent instance. (1)	27
TRYTOCOMPACT	Applicable only to the LTRA model. When specified, the simulator tries to condense LTRA transmission lines past history of input voltages and currents (1).	Not Set
TRTOL	Transient analysis error tolerance (2)	7.0
ITL1	Maximum number of iterations in computing the dc operating point (2)	100
ITL2	Maximum number of iterations in dc transfer curve analysis (2)	50
ITL4	Transient analysis timepoint iteration limit (2)	10
DEFL	Default channel length for a MOS-transistor (3)	100μm
DEFW	Default channel width for a MOS-transistor (3)	100μm

DEFAD	Default drain diffusion area for a MOS-transistor (3)	0.0
DEFAS	Default source diffusion area for a MOS-transistor (3)	0.0
PIVTOL	Minimum value for an element to be accepted as a pivot element. (4)	1.0E-13
PIVREL	The minimum relative ratio between the largest element in the column and a accepted pivot element (4)	1.0E-13
METHOD	Sets the numerical integration method used by AIM-Spice. Possible methods are Gear or Trapezoidal. (4)	Trap

The numbers in parentheses above indicates which group they belong in.

Title Line

General form:

Any text

Example:

```
SIMPLE DIFFERENTIAL PAIR
MOS OPERATIONAL AMPLIFIER
```

This must be the first line in the circuit description.

Comment line

General form:

```
* (arbitrary text)
```

Example:

```
* MAIN CIRCUIT STARTS HERE
```

An asterisk in the first column indicates that this line is a comment line. Comment lines may be placed anywhere in the circuit description.

A	Heterostructure Field Effect Transistors (HFETs)

General form:

```
AXXXXXXX ND NG NS MNAME <L=VALUE> <W=VALUE> <OFF>
+ <IC=VDS,VGS>
```

Example:

```
a1 7 2 3 hfeta l=1u w=10u
```

`ND`, `NG` and `NS` are the drain, gate and source nodes, respectively. `MNAME` is the model name, `L` is the channel length, `W` is the channel width, and `OFF` indicates a optional initial value for the element in a dc analysis. The optional initial value `IC=VDS,VGS` is meant to be used together with UIC in a transient analysis. See the description of the .IC control statement for a better way to set transient initial conditions.

HFET Model

```
.MODEL {model name} NHFET <model parameters>
```

The HFET model is a unified extrinsic model as described in section 4.6. The model parameters are listed below. Note that the following default values corresponds to a n-channel devices used as an example in section 4.6 in [5].

Name	Parameter	Units	Default
VTO	Pinch-off voltage	V	0.15
RD	Drain ohmic resistance	Ohm	0
RS	Source ohmic resistance	Ohm	0
DI	Thickness of interface layer	m	0.04e-6
LAMBDA	Output conductance parameter	1/V	0.15
VS	Saturation velocity	m/s	1.5E5
ETA	Subthreshold ideality factor	-	1.28
M	Knee shape parameter	-	3
MC	Knee shape parameter	-	3
GAMMA	Knee shape parameter	-	3
SIGMA0	DIBL parameter	-	0.057
VSIGMAT	DIBL parameter	V	0.3
VSIGMA	DIBL parameter	V	0.1
MU	Low field mobility	m^2/Vs	0.4
DELTA	Transition width parameter	-	3
VS	Saturation velocity	m/s	1.5e5
NMAX	Maximum sheet charge density in the channel	m^{-2}	2e16

DELTAD	Thickness correction	m	4.5e-9
EPSI	Dielectric constant for interface layer	F/m	1.0841E-10
JS1D	Forward gate drain diode saturation current density	A/m^2	0
JS2D	Reverse gate drain diode saturation current density	A/m^2	0
JS1S	Forward gate source diode saturation current density	A/m^2	0
JS2S	Reverse gate source diode saturation current density	A/m^2	0
M1D	Forward gate drain diode ideality factor	-	1
M2D	Reverse gate drain diode ideality factor	-	1
M1S	Forward gate source diode ideality factor	-	1
M2S	Reverse gate source diode ideality factor	-	1
RGD	Gate-drain ohmic resistance	Ω	0
RGS	Gate-source ohmic resistance	Ω	0
ALPHAG	Drain-source correction current gain	–	0

Supported Analyses

Distortion, Noise, and Pole-Zero Analysis not supported.

B Non-linear Dependent Sources

General form:

```
BXXXXXXX N+ N- <I=EXPR> <V=EXPR>
```

Example:

```
b1 0 1 i=cos(v(1))+sin(v(2))
b1 0 1 v=ln(cos(log(v(1,2)^2)))-v(3)^4+v(2)^v(1)
b1 3 4 i=17
b1 3 4 v=exp(pi^i(vdd))
```

`N+` and `N-` is the positive and negative node respectively. The values of the `V` and `I` parameters determine the voltages and currents across and through the device respectively. If `I` is given then the device is a current source, and if `V` is given the device is a voltage source. One and only one of these parameters must be given.

The small-signal AC behaviour of the non-linear source is a linear dependent source (or sources) with a proportionality constant equal to the derivative (or derivatives) of the source at the DC operating point.

The expressions given for `V` and `I` may be any function of voltages and currents through voltage sources in the system. The following functions of real variables are defined:

abs	asinh	cosh	sin
acos	atan	exp	sinh
acosh	atanh	ln	sqrt
asin	cos	log	tan

The following operations are defined:

+ − * / ^ unary −

If the argument of `log`, `ln`, or `sqrt` becomes less than zero, the absolute value of the argument is used. If a divisor becomes zero or the argument of `log` or `ln` becomes zero, an error will result. Other problems may occur when the argument for a function in a partial derivative enters a region where that function is undefined.

To get time into the expression you can integrate the current from a constant current source with a capacitor and use the resulting voltage (don't forget to set the initial voltage across the capacitor). Non-linear capacitors, resistors, and inductors may be synthesized with the non-linear dependent source. Non-linear resistors are obvious. Non-linear capacitors and inductors are implemented with their linear counterparts by a change of variables implemented with the non-linear dependent source. The following subcircuit will implement a non-linear capacitor:

```
* Bx: calculate f(input voltage)
Bx 1 0 v=f(v(pos,neg))
* Cx: linear capacitance
Cx 2 0 1
* Vx: Ammeter to measure current into the capacitor
Vx 2 1 DC 0 Volts
* Drive the current through Cx back into the circuit
Fx pos neg Vx 1
```

Non-linear inductors are similar.

Supported Analyses

All

General form:

```
CXXXXXXX N+ N- VALUE <IC=Initial values>
```

Examples:

```
cl 66 0 70pf
CBYP 17 23 10U IC=3V
```

N+ and N− are the positive and negative element nodes respectively. VALUE is the capacitance in Farads.

The optional initial value is the initial time zero value of the capacitor voltage in volts. Notice that the value is used only when the option UIC is specified in a transient analysis.

Semiconductor Capacitors

General form:

```
CXXXXXXX N1 N2 <VALUE> <MNAME> <L=LENGTH> <W=WIDTH>
+ <IC=VALUE>
```

Examples:

```
CMOD 3 7 CMODEL L=10U W=1U
```

This is a more general model for the capacitor than the one presented above. This one gives you the possibility to model temperature effects and calculation of exact capacitance based on geometric and process information. VALUE if given, defines the capacitance, and geometric information will be ignored. If MNAME is specified, the capacitance value is calculated based on information about the process in the model statement and geometric information. If VALUE is not given, then MNAME and LENGTH must be specified. If WIDTH is not given, then it will be taken form the models default width.

Capacitor Model

```
.MODEL {model name} C <model parameters>
```

The model contains process information used to calculate the capacitance value based on geometric information.

Name	Parameter	Unit	Default
CJ	Junction bottom capacitance	F/m^2	-
CJSW	Junction sidewall capacitance	F/m	-
DEFW	Default width	m	1e-6
NARROW	Narrowing due to side etching	m	0.0

The following equation is used to calculate the capacitance:

$$C = CJ \cdot (L - NARROW) \cdot (W - NARROW) + 2 \cdot CJSW \cdot (L + W - 2 \cdot NARROW)$$

Supported Analyses

All

General form:

```
DXXXXXXX N+ N- MNAME <AREA> <OFF> <IC=VD> <TEMP=T>
```

Examples:

```
DBRIDGE 2 10 DIODE1
DCLMP 3 7 DMOD 3.0 IC=0.2
```

N+ and N− are the positive and negative nodes, respectively. MNAME is the model name, AREA is the area factor, and OFF indicates an optional initial value during a dc analysis. If the area factor is not given, 1 is assumed. The optional initial value IC=VD is meant to be used together with an UIC in a transient analysis. The optional TEMP value is the temperature at which this device is to operate, and overrides the temperature specified as an option.

Diode Model

```
.MODEL {model name} D <model parameters>
```

In AIM-Spice there are 2 diode models. Level 1 is the standard diode model supported from Berkeley, which is the default model. Level 2 is a GaAs heterostructure diode model described in section 1.10 in [5]. To select the heterostructure diode model specify LEVEL=2 on the model line.

Level 1 model parameters are:

Name	Parameter	Units	Default
IS	Saturation current (level 1 only)	A	1.0e-14
RS	Ohmic resistance	Ohm	0
N	Emission coefficient	-	1
TT	Transit time	s	0
CJO	Zero bias junction capacitance	F	0
VJ	Junction potential	V	1
M	Grading coefficient	-	0.5
EG	Activation energy	eV	1.11
XTI	Saturation current temperature exponent	-	3.0
KF	Flicker noise coefficient	-	0
AF	Flicker noise exponent	-	1
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5
BV	Reverse breakdown voltage	V	infinite

IBV	Current at breakdown voltage	A	1.0e-3
TNOM	Parameter measurement temperature	°C	27

Level 2 model parameters are (in addition to those for level 1):

Name	**Parameter**	**Units**	**Default**
DN	Diffusion constant for electrons	m^2/s	0.02
DP	Diffusion constant for holes	m^2/s	0.000942
LN	Diffusion length for electrons	m	7.21e-5
LP	Diffusion length for holes	m	8.681e-7
ND	Donor doping density	m^{-3}	7.0e24
NA	Acceptor doping density	m^{-3}	3e22
DELTAEC	Conduction band discontinuity	eV	0.6
XP	p-region width	m	1μm
XN	n-region width	m	1μm
EPSP	Dielectric constant on p-side	F/m	1.0593e-10
EPSN	Dielectric constant on n-side	F/m	1.1594e-10

Temperature Effects

Temperature appears explicitly in the exponential terms.

Temperature dependence of the saturation current in the junction diode model is determined by:

$$I_S(T_1) = I_S(T_0)\left(\frac{T_1}{T_0}\right)^{\frac{XTI}{N}} \exp\left(\frac{E_g q T_1 T_0}{Nk(T_1 - T_0)}\right)$$

where k is Boltzmann's constant, q is the electronic charge, E_g is the energy gap, XTI is the saturation current temperature exponent, and N is the emission coefficient. The last three quantities are model parameters.

Note that for Schottky barrier diodes, the value for XTI is usually 2.

Supported Analyses

All

E Linear Voltage-Controlled Voltage Sources

General form:

```
EXXXXXXX N+ N- NC+ NC- VALUE
```

Example:

```
E1 2 3 14 1 2.0
```

N+ and N− are the positive and negative nodes, respectively. NC+ and NC− are the positive and negative controlling nodes, respectively. VALUE is the voltage gain.

Supported Analyses

All

F Linear Current-Controlled Current Sources

General form:

```
FXXXXXXX N+ N− VNAME VALUE
```

Example:

```
F1 14 7 VIN 5
```

`N+` and `N−` are the positive and negative nodes, respectively. Current flows from the positive node through the source to the negative node. `VNAME` is the name of the voltage source where the controlling current flows. The direction of positive control current is from positive node through the source to the negative node of `VNAME`. `VALUE` is the current gain.

Supported Analyses

All

G Linear Voltage-Controlled Current Sources

General form:

```
GXXXXXXX N+ N- NC+ NC- VALUE
```

Example:

```
G1 2 0 5 0 0.1MMHO
```

`N+` and `N−` are the positive and negative nodes, respectively. Current flows from the positive node through the source to the negative node. `NC+` and `NC−` are the positive and negative controlling nodes, respectively. `VALUE` is the transconductance in mhos.

Supported Analyses

All

H Linear Current-Controlled Voltage Sources

General form:

```
HXXXXXXX N+ N- VNAME VALUE
```

Example:

```
HX1 6 2 Vz 0.5K
```

`N+` and `N-` are the positive and negative nodes, respectively. Current flows from the positive node through the source to the negative node. `VNAME` is the name of the voltage source where the controlling current flows. The direction of positive control current is from positive node through the source to the negative node of `VNAME`. `VALUE` is the transresistance in ohm.

Supported Analyses

All

I Independent Current Sources

General form:

```
IYYYYYYY N+ N- <<DC> DC/TRAN VALUE> <AC <ACMAG <ACPHASE>>>
+ <DISTOF1 <F1MAG <F1PHASE>>> <DISTOF2 <F2MAG <F2PHASE>>>
```

Examples:

```
isrc 23 21 ac 0.333 45.0 sffm(0 1 10k 5 1k)
```

`N+` and `N−` are the positive and negative nodes, respectively. Positive current flows from the positive node through the source to the negative node.

`DC/TRAN` is the source value during a dc or a transient analysis. The value can be omitted if it is zero for both the dc and transient analysis. If the source is time invariant, its value can be prefixed with `DC`.

`ACMAG` is amplitude value and `ACPHASE` is the phase value of the source during an ac analysis. If `ACMAG` is omitted after the keyword `AC`, 1 is assumed. If `ACPHASE` is omitted, 0 is assumed.

`DISTOF1` and `DISTOF2` are the keywords that specify that the independent source has distortion inputs at the frequencies `F1` and `F2` respectively (see the description of the distortion analysis parameters). The keywords may be followed by an optional magnitude and phase. The default values of the magnitude and phase are 1.0 and 0.0 respectively.

All independent sources can be assigned time varying values during a transient analysis. If a source is assigned a time varying value, the value for t=0 is used during a dc analysis. There are 5 predefined functions for time varying sources: pulse, exponent, sinus, piece-wise linear, and single frequency FM. If parameters are omitted, default values shown below will be assumed. In the tables below is DT and T2 increment time and final time in a transient analysis, respectively[1].

Pulse

General form:

```
PULSE(I1 I2 TD TR TF PW PER)
```

Parameters	Default values	Units
I1 (initial value)	None	Amp
I2 (pulsed value)	None	Amp

[1] See the section describing the transient analysis

TD (delay time)	0.0	seconds
TR (rise time)	DT	seconds
TF (fall time)	DT	seconds
PW (pulse width)	T2	seconds
PER (period)	T2	seconds

Example:

```
IB 3 0 PULSE(1 5 1S 0.1S 0.4S 0.5S 2S)
```



Sinus

General form:

```
SIN(I0 IA FREQ TD THETA)
```

Parameters	**Default values**	**Units**
I0 (offset)	None	Amp
IA (amplitude)	None	Amp
FREQ (frequency)	1/T2	Hz
TD (delay)	0.0	seconds
THETA(damping factor)	0.0	1/seconds

The shape of the waveform is:

<u>0 < time < TD</u>

I = *I0*

TD < time < T2

$$I = I0 + IA\sin\left(2\pi \cdot FREQ \cdot (time + TD)\right) \cdot \exp\left(-(time - TD) \cdot THETA\right)$$

Example:

```
IB 3 0 SIN(2 2 5 1S 1)
```



Exponent

General form:

```
EXP(I1 I2 TD1 TAU1 TD2 TAU2)
```

Parameters	Default values	Units
I1 (initial value)	None	Amp
IA (pulsed value)	None	Amp
TD1(rise delay time)	0.0	seconds
TAU1(rise time constant)	DT	seconds
TD2 (delay fall time)	TD1+DT	seconds
TAU2 (fall time constant)	DT	seconds

The shape of the waveform is:

0 < time < TD1

$I = I1$

TD1 < time < TD2

$$I = I1 + (I2 - I1) \cdot \left(1 - \exp\left(-(time - TD1) \cdot TAU1\right)\right)$$

TD2 < time < T2

$$I = I1 + (I2 - I1) \cdot (1 - \exp(-(time - TD1) \cdot TAU1)) + (I1 - I2) \cdot (1 - \exp(-(time - TD2) \cdot TAU2))$$

Example:

```
IB 3 0 EXP(1 5 1S 0.2S 2S 0.5S)
```



Piece-wise Linear

General form:

```
PWL(T1 I1 <T2 I2 T3 I3 T4 I4 T5 I5 ....>)
```

Parameters and default values:

Every pair of values (T_i, I_i) specifies that the value of the source is I_i at T_i. The value of the source between these values is calculated using a linear interpolation with the given values.

Example:

```
ICLOCK 7 5 PWL(0 0 1 0 1.2 4 1.6 2.0 2.0 5.0 3.0 1.0)
```



Single frequency FM

General form:

```
SFFM(I0 IA FC MDI FS)
```

Parameters	Default values	Units
I0 (offset)	None	Amp
IA (amplitude)	None	Amp
FC (carrier frequency)	1/T2	Hz
MDI (modulation index)	None	-
FS (signal frequency)	1/T2	Hz

The shape of the waveform is:

$$I = I0 + IA \cdot \sin\left(\left(2\pi \cdot FC \cdot time\right) + MDI \cdot \sin\left(2\pi \cdot FS \cdot time\right)\right)$$

Example:

```
IB 12 0 SFFM(0 1M 20K 5 1K)
```



Supported Analyses

All

General form:

```
JXXXXXXX ND NG NS MNAME <AREA> <OFF> <IC=VDS,VGS> <TEMP=T>
```

Example:

```
J1 7 2 3 JM1 OFF
```

ND, NG and NS are the drain, gate and source nodes, respectively. MNAME is the model name, AREA is the area factor, and OFF indicates a optional initial value for the element in a dc analysis. If the area factor is omitted, 1.0 is assumed. The optional initial value IC=VDS,VGS is meant to be used together with UIC in a transient analysis. See the description of the .IC control statement for a better way to set transient initial conditions. The optional TEMP value is the temperature at which this device is to operate, and overrides the temperature specified in the option value.

JFET Model

```
.MODEL {model name} NJF <model parameters>
.MODEL {model name} PJF <model parameters>
```

Name	Parameter	Units	Default
VTO	Threshold voltage	V	-2.0
BETA	Transconductance parameter	A/V^2	1.0e-4
LAMBDA	Channel length modulation parameter	1/V	0
RD	Drain resistance	Ohm	0
RS	Source resistance	Ohm	0
CGS	Zero-bias G-S junction capacitance	F	0
CGD	Zero-bias G-D junction capacitance	F	0
PB	Gate junction potential	V	1
IS	Gate junction saturation current	A	1.0E-14
KF	Flicker noise coefficient	-	0
AF	Flicker noise exponent	-	1
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5
TNOM	Parameter measurement temperature	°C	27

Temperature Effects

Temperature appears explicitly in the exponential terms.

Temperature dependence of the saturation current in the two gate junctions of the model is determined by:

$$I_S(T_1) = I_S(T_0)\exp\left[1.11\left(\frac{T_1}{T_0} - 1\right) / V_{th}\right]$$

where V_{th} is the thermal voltage.

Supported Analyses

All

K	Coupled Inductors (transformers)

General form:

```
KXXXXXXX LYYYYYYY LZZZZZZZ VALUE
```

Examples:

```
k43 laa lbb 0.9999
kxfrmr l1 l2 0.82
```

`LYYYYYYY` and `LZZZZZZZ` are the names of the two coupled inductors, and `VALUE` is the coefficient of coupling, K, which must be greater than 0 and less than or equal to 1. Using the dot convention, place a dot on the first node of each inductor.

The relation between the coupling coefficient, K, and the mutual inductance is given by

$$K = \frac{M_{ij}}{\sqrt{L_i L_j}},$$

where L_i and L_j are a coupled pair of inductors, and M_{ij} is the mutual inductance between L_i and L_j.

Supported Analyses

All

General form:

```
LYYYYYYY N+ N- VALUE <IC=Initial values>
```

Examples:

```
llink 42 69 1uh
lshunt 23 51 10u ic=15.7ma
```

N+ and N− are the positive and negative element nodes respectively. VALUE is the inductance in Henries.

The optional initial value is the initial time zero value of the inductor current in amps that flows from N+ through the inductor to N−. Notice that the value is used only when the option UIC is specified in a transient analysis.

Supported Analyses

All

M Metal Oxide Semiconductor Field Effect Transistors (MOSFETs)

General form:

```
MXXXXXXX ND NG NS NB MNAME <L=VALUE> <W=VALUE> <AD=VALUE>
+ <AS=VALUE> <PD=VALUE> <PS=VALUE> <NRD=VALUE>
+ <NRS=VALUE> <OFF> <IC=VDS,VGS,VBS> <TEMP=T>
```

Example:

```
M1 24 2 0 20 TYPE1
m15 15 15 12 32 m w=12.7u  l=207.8u
M1 2 9 3 0 MOD1 L=10U W=5U AD=100P AS=100P PD=40U PS=40U
```

`ND`, `NG` and `NS` are the drain, gate, source and bulk (substrate) nodes, respectively. `MNAME` is the model name, `L` and `W` is the channel length and width in meters, respectively. `AD` and `AS` is the drain and source diffusion areas in square meters. If any of `L`, `W`, `AD` or `AS` is not given, default values is used. `PD` and `PS` is the perimeters of the drain and source diffusion areas. `NRD` and `NRS` is the relative resistivities for drain and source in number of squares, respectively. Default values for `PD` and `PS` are 0.0, while default values for `NRD` and `NRS` are 1.0. `OFF` indicates a optional initial value for the element in a dc analysis. The optional initial value `IC=VDS,VGS,VBS` is meant to be used together with UIC in a transient analysis. See the description of the .IC control statement for a better way to set transient initial conditions. The optional `TEMP` value is the temperature at which this device is to operate, and overrides the temperature specified in the option value.

Note! The substrate node is ignored in level 11 and 12, and so is the VBS initial voltage.

MOSFET Model

```
.MODEL {model name} NMOS <model parameters>
.MODEL {model name} PMOS <model parameters>
```

In AIM-Spice there are 12 MOSFET models. The parameter LEVEL selects which model to use.

LEVEL=1	Shichman-Hodges
LEVEL=2	Geometric based analytical model
LEVEL=3	Semi-empirical short channel model
LEVEL=4	BSIM (Berkeley Short Channel Igfet Model)
LEVEL=5	New BSIM (BSIM2 as described in [3])
LEVEL=6	MOS6 (as described in [4])

LEVEL=7	An universal extrinsic short channel MOS model (as described in Section 3.9 in [5])
LEVEL=8	An unified long channel MOS model (as described in Section 3.10 and 3.11 in [5])
LEVEL=9	A short channel MOS model (as described in Section 3.10 and 3.11 in [5])
LEVEL=10	An unified intrinsic short channel model (as described in Section 3.10 and 3.11 in [5])
LEVEL=11	An unified extrinsic amorphous silicon thin film transistor model (as described in Section 5.2 in [5])
LEVEL=12	A model for polysilicon thin film transistors (as described in Section 5.3 in [5])

Effects due to charge storage is based on a model by Meyer for level 1, 2 3, 6, 7, 8, 9, 10, 11 and 12. The BSIM models (level 4 and 5) have their own charge storage model.

Effects due to the thin-oxide capacitance is treated slightly different in level 1. Voltage dependent capacitances are included only if TOX is specified.

There is redundancy in specifying junction parameters. For example the reverse current can be specified either with the IS parameter (in Amp) or with JS (in Amp/m^2). While the first one is an absolute value, the other one is multiplied with AD and AS to give reverse current at the drain and source junctions, respectively. The last approach is preferred. The same is also true for the parameters CBD, CBS and CJ. Parasitic resistances can be given with RD and RS (in ohms) or with RSH (in ohms/square). The last one is multiplied with number of squares NRD and NRS.

Parameters for level 1, 2, 3 and 6:

Name	Parameter	Units	Default
VTO	Zero-bias threshold voltage	V	0.0
KP	Transconductance parameter	A/V^2	2.0e-5
GAMMA	Bulk threshold parameter	$\sqrt{V}$	0.0
PHI	Surface potential	V	0.6
LAMBDA	Channel length modulation (Only MOS1 and MOS2)	1/V	0.0
RD	Drain resistance	Ohm	0.0
RS	Source resistance	Ohm	0.0
CBD	Zero-bias B-D junction capacitance	F	0.0
CBS	Zero-bias B-S junction capacitance	F	0.0
IS	Bulk junction saturation current	A	1.0e-14
PB	Bulk junction potential	V	0.8

CGSO	Gate-source overlap capacitance per meter channel width	F/m	0.0
CGDO	Gate-drain overlap capacitance per meter channel width	F/m	0.0
CGBO	Gate-bulk overlap capacitance per meter channel width	F/m	0.0
RSH	Drain and source diffusion sheet resistance	Ohm/☐	0.0
CJ	Zero-bias bulk junction bottom capacitance per square-meter of junction area	F/m^2	0.0
MJ	Bulk junction bottom grading coefficient	-	0.5
CJSW	Zero-bias bulk junction sidewall capacitance per meter of junction perimeter	F/m	0.0
MJSW	Bulk junction sidewall grading coefficient	-	0.50 (level 1) 0.33 (level 2)
JS	Bulk junction saturation current per m^2 of junction area	A/m^2	0
TOX	Thin-oxide thickness	m	1.0e-7
NSUB	Substrate doping	$1/cm^3$	0.0
NSS	Surface state density	$1/cm^2$	0.0
NFS	Fast surface state density	$1/cm^2$	0.0
TPG	Type of gate material: +1 : opposite of substrate -1 : same as substrate 0 : Al gate	-	1.0
XJ	Metallurgical junction depth	m	0.0
LD	Lateral diffusion	m	0.0
U0	Surface mobility	cm^2/Vs	600
UCRIT	Critical field for mobility degradation (Only MOS2)	V/cm	1.0e4
UEXP	Critical field exponent in mobility degradation (Only MOS2)	-	0.0
UTRA	Transverse field coefficient (deleted for MOS2)	-	0.0
VMAX	Maximum drift velocity for carriers	m/s	0.0
NEFF	Total channel charge (fixed and mobile) coefficient (Only MOS2)	-	1.0
KF	Flicker noise coefficient	-	0.0
AF	Flicker noise exponent	-	1.0
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5

DELTA	Width effect on threshold voltage (Only MOS2 and MOS3)	-	0.0
THETA	Mobility modulation (MOS3 only)	1/V	0.0
ETA	Static feedback (Only MOS3)	-	0.0
KAPPA	Saturation field factor (MOS3 only)	-	0.2
TNOM	Parameter measurement temperature	°C	27

MOSFET level 4 and 5 is BSIM (Berkeley Short Channel IGFET Model) models. Parameters for these models are obtained from process characterization, and they can be automatically generated.

Parameters marked with an '*' in the l/w column below have length and width dependency. For example VFB is the basic parameter with units Volt, and LVFB and WVFB also exist and have units Volt · µm. The equation

$$P = P_0 + \frac{P_L}{L_{effective}} + \frac{P_W}{W_{effective}}$$

are used to compute the parameter for the actual element given by

$$L_{effective} = L_{input} - DL$$

and

$$W_{effective} = W_{input} - DW$$

Note that BSIM models are meant to be used together with a process characterization system. None of the parameters in BSIM models have default values, and leaving one out is considered as an error.

AIM-Spice BSIM (level 4 and 5) parameters:

Name	**Parameter**	**Units**	**l/w**
VFB	Flat band voltage	V	*
PHI	Surface inversion potential	V	*
K1	Body effect coefficient	$V^{1/2}$	*
K2	Drain/source depletion charge sharing coefficient	-	*
ETA	Zero-bias drain-induced barrier lowering coefficient	-	*
MUZ	Zero-bias mobility	cm^2/Vs	
DL	Shortening of channel	µm	
DW	Narrowing of channel	µm	
U0	Zero-bias transverse-field mobility degradation coefficient		*
U1	Zero-bias velocity saturation coefficient	µm/V	*

X2MZ	Sens. of mobility to substrate bias at V_{ds}=0	cm^2/V^2s	*
X2E	Sens. of drain-induced barrier lowering effect to substrate bias		*
X3E	Sens. of drain-induced barrier lowering effect to drain bias at $V_{ds}= V_{dd}$		*
X2U0	Sens. of transverse field mobility degradation effect to substrate bias	$1/V^2$	*
X2U1	Sens. of velocity saturation effect to substrate bias	μmV^{-2}	*
MUS	Mobility at zero substrate bias and at $V_{ds}= V_{dd}$	cm^2/V^2s	
X2MS	Sens. of mobility to substrate bias at $V_{ds}= V_{dd}$	cm^2/V^2s	*
X3MS	Sens. of mobility to drain bias at $V_{ds}= V_{dd}$	μmV^{-2}	*
X3U1	Sens. of velocity saturation effect on drain bias at $V_{ds}= V_{dd}$	μmV^{-2}	*
TOX	Gate oxide thickness	µm	
TEMP	Temperature at which parameters were measured	C	
VDD	Measurement bias range	V	
CGDO	Gate-drain overlap capacitance per meter channel width	F/m	
CGSO	Gate-source overlap capacitance per meter channel width	F/m	
CGBO	Gate-bulk overlap capacitance per meter channel width	F/m	
XPART	Gate-oxide capacitance charge model flag	-	
N0	Zero-bias subthreshold slope coefficient	-	*
NB	Sens. of subthreshold slope to substrate bias	-	*
ND	Sens. of subthreshold slope to drain	-	*
RSH	Drain and source diffusion sheet resistance	Ohm/☐	
JS	Source drain junction current density	A/m^2	
PB	Built in potential of source drain junction	V	
MJ	Grading coefficient of source drain junction	-	
PBSW	Built in potential of source drain junction sidewall	V	
MJSW	Grading coefficient of source drain junction sidewall	-	
CJ	Source drain junction capacitance per unit area	F/m^2	
CJSW	Source drain junction sidewall capacitance per unit length	F/m	
WDF	Source drain junction default width	m	
DELL	Source drain junction length reduction	m	

XPART=0 selects a 40/60 drain/source partition in saturation, while XPART=1 selects 0/100 drain/source charge partition.

AIM-Spice model parameters common for level 7, 8, 9, and 10 are listed below.

Name	Parameter	Units	Default
VTO	Zero-bias threshold voltage	V	0.0
GAMMA	Bulk threshold parameter	$\sqrt{V}$	0.0
PHI	Surface potential	V	0.6
RD	Drain resistance	Ohm	0.0
RS	Source resistance	Ohm	0.0
CBD	Zero-bias B-D junction capacitance	F	0.0
CBS	Zero-bias B-S junction capacitance	F	0.0
IS	Bulk junction saturation current	A	1.0e-14
PB	Bulk junction potential	V	0.8
CGSO	Gate-source overlap capacitance per meter channel width	F/m	0.0
CGDO	Gate-drain overlap capacitance per meter channel width	F/m	0.0
CGBO	Gate-bulk overlap capacitance per meter channel width	F/m	0.0
RSH	Drain and source diffusion sheet resistance	Ohm/□	0.0
CJ	Zero-bias bulk junction bottom capacitance per square-meter of junction area	F/m^2	0.0
MJ	Bulk junction bottom grading coefficient	-	0.5
CJSW	Zero-bias bulk junction sidewall capacitance per meter of junction perimeter	F/m	0.0
MJSW	Bulk junction sidewall grading coefficient	-	0.33
JS	Bulk junction saturation current per m^2 of junction area	A/m^2	0
TOX	Thin-oxide thickness	m	1.0e-7
NSUB	Substrate doping	1/cm^3	0.0
NSS	Surface state density	1/cm^2	0.0
TPG	Type of gate material: +1 : opposite of substrate -1 : same as substrate 0 : Al gate	-	1.0
XJ	Metallurgical junction depth	m	0.0
LD	Lateral diffusion	m	0.0
U0	Surface mobility	cm^2/Vs	600

VMAX	Maximum drift velocity for carriers	m/s	4.0e4 (level 7) 6.0e4 (level 8,9) 8.6e4 (level 10)
KF	Flicker noise coefficient	-	0.0
AF	Flicker noise exponent	-	1.0
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5
DELTA	Width effect on threshold voltage	-	0.0
TNOM	Parameter measurement temperature	°C	27

Model parameters local to level 7 model are:

Name	**Parameter**	**Units**	**Default (NMOS)**
LAMBDA	Transconductance parameter	V^{-1}	0.048
ETA	Subthreshold ideality factor	-	1.32
M	Knee shape parameter	-	4.0
SIGMA0	DIBL parameter	-	0.048
VSIGMAT	DIBL parameter	V	1.7
VSIGMA	DIBL parameter	V	0.2

Model parameters local to level 8 model are:

Name	**Parameter**	**Units**	**Default**
ALPHA	Parameter accounting for the threshold dependence on the channel potential	-	1.164 (NMOS) 1.4 (PMOS)
K1	Mobility parameter	cm^2/V^2s	28 (NMOS) 1510 (PMOS)
ETA	Subthreshold ideality factor	-	1.3 (NMOS) 1.2 (PMOS)

Model parameters local to level 9 model are:

Name	**Parameter**	**Units**	**Default**
ALPHA	Parameter accounting for the threshold dependence on the channel potential	-	1.2 (NMOS) 1.34 (PMOS)
XI	Saturation voltage parameter (NMOS only)	-	0.79
GAM	Saturation point parameter	-	3.0 (NMOS) 2.35 (PMOS)

K1	Mobility parameter	cm^2/V^2s	28 (NMOS) 1510 (PMOS)
LAMBDA	Characteristic length of the saturated region of the channel	m	994Å (NMOS) 1043Å (PMOS)
ZETA	Velocity saturation factor (PMOS only)	-	0.34

Model parameters local to level 10 model are:

Name	Parameter	Units	Default
ALPHA	Bulk charge factor	-	1.05 (NMOS) 1.3 (PMOS)
ETA	Ideality factor in subthreshold	-	1.42 (NMOS) 1.39 (PMOS)
SIGMA	Parameter accounting for DIBL effects	-	0.03 (NMOS) 0.007 (PMOS)
K1	Mobility parameter	cm^2/V^2s	28 (NMOS) 1510 (PMOS)
LAMBDA	Channel length modulation parameter	m	8.58E-8 (NMOS) 7.25E-8 (PMOS)
XI	Saturation voltage factor (NMOS only)	-	0.76
ZETA	Velocity saturation factor (PMOS only)	-	0.28
GAM	Saturation point parameter	-	3.0 (NMOS) 2.35 (PMOS)

Model parameters common to level 11 and 12 are listed below.

Name	Parameter	Units	Default
VTO	Zero-bias threshold voltage	V	0.0
GAMMA	Bulk threshold parameter	$\sqrt{V}$	0.0
PHI	Surface potential	V	0.6
RD	Drain resistance	Ohm	0.0
RS	Source resistance	Ohm	0.0
CGSO	Gate-source overlap capacitance per meter channel width	F/m	0.0
CGDO	Gate-drain overlap capacitance per meter channel width	F/m	0.0
RSH	Drain and source diffusion sheet resistance	Ohm/□	0.0
TOX	Thin-oxide thickness	m	1.0e-7
NSUB	Substrate doping	$1/cm^3$	0.0

NSS	Surface state density	$1/cm^2$	0.0
TPG	Type of gate material: +1 : opposite of substrate -1 : same as substrate 0 : Al gate	-	1.0
LD	Lateral diffusion	m	0.0
U0	Surface mobility	cm^2/Vs	600
TNOM	Parameter measurement temperature	°C	27

Note! The following columns with default model parameters corresponds to n-channel TFT devices used as examples in [5].

Model parameters local to level 11 model are:

Name	Parameter	Units	Default
M1	Knee shape parameter	-	2.5
M2	Exponent in mobility expressions	-	2.5
E2	Characteristic energy	J	$k_B \cdot 300.15$
VSIGMA	Parameter accounting for DIBL effects	V	0.2
VSIGMAT	Parameter accounting for DIBL effects	V	1.7
SIGMA0	Parameter accounting for DIBL effects	-	0.048
LAMBDA	Transconductance parameter	1/V	0.0048
N0	Scaling factor	m^{-3}	1E18
ASUB	Fitting parameter	J/V	E2/5

Model parameters local to level 12 model are:

Name	Parameter	Units	Default
V0	Scaling voltage	V	10.7
GAM	Saturation voltage parameter	-	0.03
ETA	Ideality factor in subthreshold regime	-	6.9

Temperature Effects

Temperature appears explicitly in the exponential terms in the equations describing current across the bulk junctions.

Temperature appears explicitly in the value of junction potential, ϕ (in AIM-Spice PHI). The temperature dependence is determined by:

$$\phi(T) = \frac{kT}{q} \log_e \left(\frac{N_a N_d}{N_i(T)^2} \right)$$

where k is Boltzmann's constant, q is the electronic charge, N_a is the acceptor impurity density, N_d is the donor impurity density, and N_i is the intrinsic carrier concentration.

Temperature appears explicitly in the value of surface mobility μ_0 (or U0). The temperature dependence is determined by:

$$\mu_0(T) = \frac{\mu_0(T_0)}{\left(T / T_0\right)^{1.5}}$$

Supported Analyses

Level 1, 2, 3:	All
Level 4:	Noise Analysis not supported
Level 5:	Distortion and Noise Analysis not supported
Level 6:	AC, Distortion, Noise, and Pole-Zero Analysis not supported
Level 7, 8, 9, 10, 11, 12:	Distortion and Noise Analysis not supported

General form:

```
NXXXXXXX NC NB NE <NS> MNAME <AREA> <OFF> <IC=VBE,VCE>
+ <TEMP=T>
```

Example:

```
N23 10 24 13 NMOD IC=0.6,5.0
n2 5 4 0 nnd
```

NC, NB and NE are the collector, base and emitter nodes, respectively. NS is the substrate node. If this is not given, ground is assumed. MNAME is the model name, AREA is the area factor, and OFF indicates a optional initial value for the element in a dc analysis. If the area factor is omitted, 1.0 is assumed. The optional initial value IC=VBE,VCE is meant to be used together with UIC in a transient analysis. See the description of the .IC control statement for a better way to set transient initial conditions. The optional TEMP value is the temperature at which this device is to operate, and overrides the temperature specified in the option value.

HBT Model

```
.MODEL {model name} HNPN <model parameters>
.MODEL {model name} HPNP <model parameters>
```

The heterojunction bipolar transistor model in AIM-Spice is a modification of the Ebers-Moll bipolar transistor model.

Name	Parameter	Units	Default
IS	Transport saturation current	A	1e-16
BF	Ideal maximum forward beta	-	100
NF	Forward current emission coefficient	-	1.0
ISE	B-E leakage saturation current	A	0
NE	B-E leakage emission coefficient	-	1.2
BR	Ideal maximum reverse beta	-	1
NR	Reverse current emission coefficient	-	1
ISC	B-C leakage saturation current	A	0
NC	B-C leakage emission coefficient	-	2
RB	Base resistance	Ohm	0
RE	Emitter resistance	Ohm	0
RC	Collector resistance	Ohm	0
CJE	B-E zero bias depletion capacitance	F	0
VJE	B-E built-in potential	V	0.75

MJE	B-E junction exponential factor	-	0.33
TF	Ideal forward transit time	s	0
XTF	Coefficient for bias dependence of TF	-	0
VTF	Voltage describing VBC dependence of TF	V	infinite
ITF	High current parameter for effect on TF	A	0
PTF	Excess phase at f=1.0/(TF · 2π) Hz	Deg	
CJC	B-C zero bias depletion capacitance	F	0
VJC	B-C built-in potential	Volt	0.75
MJC	B-C junction exponential factor	-	0.33
XCJC	Fraction of B-C depletion capacitance connected to internal base node.	-	1
TR	Ideal reverse transit-time	s	0
CJS	zero-bias collector-substrate capacitance	F	0
VJS	Substrate junction built-in potential	V	0.75
MJS	Substrate junction exponential factor	-	0
XTB	Forward and reverse beta temperature exponent	-	0
EG	Energy gap for temperature effect on IS	eV	1.11 (Si)
XTI	Temperature exponent for effect on IS	-	3
KF	Flicker-noise coefficient	-	0
AF	Flicker-noise exponent	-	1
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5
TNOM	Parameter measurement temperature	°C	27
IRB0	Base region recombination saturation current	A	0
IRS1	Surface recombination saturation current 1	A	0
IRS2	Surface recombination saturation current 2	A	0
ICSAT	Collector saturation current	A	0
M	Knee shape parameter	-	3

The modifications to the Ebers-Moll model are two new contributions to the recombination current and a new expression for β_F.

Recombination in the base region is modelled by the expression

$$I_{rb} = IRB0\left(\mathrm{e}^{V_{be}/V_{th}} - 1\right)$$

Surface recombination is modelled by the expression.

$$I_{rs} = IRS1\left(\mathrm{e}^{V_{be}/V_{th}} - 1\right) + IRS2\left(\mathrm{e}^{V_{be}/2V_{th}} - 1\right)$$

Where IRS1 and IRS2 are proportional to the emitter perimeter.

The expression for β_F is given below.

$$\beta_F = \frac{\beta_{F0}}{\left[1 + \left(I_c / ICSAT\right)^M\right]^{1/M}}$$

Supported Analyses

Distortion, Noise, and Pole-Zero Analysis not supported.

General form:

```
OXXXXXXX N1 N2 N3 N4 MNAME
```

Example:

```
o23 1 0 2 0 lmod
ocon 10 5 20 5 interconnect
```

This is a two-port convolution model for single-conductor lossy transmission lines. `N1` and `N2` are the nodes at port 1, `N3` and `N4` are the nodes at port 2. Note that a lossy transmission line with zero loss may be more accurate than the lossless transmission line due to implementation details.

LTRA Model

```
.MODEL {model name} LTRA <model parameters>
```

The uniform RLC/RC/LC/RG transmission line model (referred to as the LTRA model henceforth) models a uniform constant-parameter distributed transmission line. The RC and LC cases may also be modelled using the URC and TRA models; however, the newer LTRA model is usually faster and more accurate than the others. The operation of the LTRA model is based on the convolution of the transmission line's impulse response with its inputs see [2].

The LTRA model takes a number of parameters , some which must be given and some which are optional.

Name	Parameter	Units	Default
R	Resistance/Length	Ω/m	0.0
L	Inductance/Length	H/m	0.0
C	Capacitance/Length	F/m	0.0
G	Conductance/Length	Ω^{-1}/m	0.0
LEN	Length of line	m	-
REL	Breakpoint control	-	1
ABS	Breakpoint control	-	1
NOSTEPLIMIT	Don't limit timestep to less than line delay	Flag	not set
NOCONTROL	Don't do complex timestep control	Flag	not set
LININTERP	Use linear interpolation	Flag	not set
MIXEDINTERP	Use linear when quadratic seems bad	Flag	not set

COMPACTREL	Special reltol for history compaction		RELTOL
COMPACTABS	Special abstol for history compaction		ABSTOL
TRUNCNR	Use Newton-Raphson method for timestep control	Flag	not set
TRUNCDONTCUT	Don't limit timestep to keep impulse-response errors low	Flag	not set

The following types of lines have been implemented so far: RLC (uniform transmission line with series loss only), RC (uniform RC line), LC (lossless transmission line), and RG (distributed series resistance and parallel conductance only). Any other combination will yield erroneous results and should not be tried. The length (LEN) of line must be specified.

NOSTEPLIMIT is a flag that will remove the default restrictions of limiting time-step to less than the line delay in the RLC case. NOCONTROL is a flag that prevents the default limiting of the time-step based on convolution error criteria in the RLC and RC cases. This speeds up simulation but may in some cases reduce the accuracy of results. LININTERP is a flag that, when set, will use linear interpolation instead of the default quadratic interpolation for calculating delayed signals. MIXEDINTERP is a flag that, when set, uses a metric for judging whether quadratic interpolation is not applicable and if so uses linear interpolation. Otherwise it uses the default quadratic interpolation. TRUNCDONTCUT is a flag that removes the default cutting of the time-step to limit errors in the actual calculation of impulse-response related quantities. COMPACTREL and COMPACTABS are quantities that control the compaction of the past history of values stored for convolution. Larger values of these lower accuracy but usually increase simulation speed. These are to be used with the TRYTOCOMPACT option, described in the section about options. TRUNCNR is a flag that turns on the use of Newton-Raphson iterations to determine an appropriate timestep in the timestep control routines. The default is a trial and error procedure by cutting the previous timestep in half. REL and ABS are quantities that control the setting of breakpoints.

The option most worth experimenting with for increasing the speed of simulation is REL. The default value of 1 is usually safe from the point of view of accuracy but occasionally increases computation time. A value grater than 2 eliminates all breakpoints and may be worth trying depending on the nature of the rest of the circuit, keeping in mind that it might not be safe from the viewpoint of accuracy. Breakpoints may usually be entirely eliminated if it is expected that the circuit will not display sharp discontinuities. Values between 0 and 1 are usually not required but may be used for setting many breakpoints.

COMPACTREL may also be experimented with when the option TRYTOCOMPACT is specified. The legal range is between 0 and 1. Larger values usually decrease the accuracy of the simulation but in some cases improve speed. If TRYTOCOMPACT is not specified history compaction is not attempted and accuracy is high. NOCONTROL, TRUNCDONTCUT and NOSTEPLIMIT also tend to increase speed at the expense of accuracy.

Supported Analyses

Distortion, Noise, and Pole-Zero Analysis not supported

General form:

```
QXXXXXXX NC NB NE <NS> MNAME <AREA> <OFF> <IC=VBE,VCE>
+ <TEMP=T>
```

Example:

```
Q23 10 24 13 QMOD IC=0.6,5.0
q2 5 4 0 qnd
```

NC, NB and NE are the collector, base and emitter nodes, respectively. NS is the substrate node. If this is not given, ground is assumed. MNAME is the model name, AREA is the area factor, and OFF indicates a optional initial value for the element in a dc analysis. If the area factor is omitted, 1.0 is assumed. The optional initial value IC=VBE,VCE is meant to be used together with UIC in a transient analysis. See the description of the .IC control statement for a better way to set transient initial conditions. The optional TEMP value is the temperature at which this device is to operate, and overrides the temperature specified as a option.

BJT Model

```
.MODEL {model name} NPN <model parameters>
.MODEL {model name} PNP <model parameters>
```

The bipolar transistor model in AIM-Spice is an adaptation of the Gummel Poon model. In AIM-Spice the model is extended to include high bias effects. The model automatically simplifies to Ebers Moll if certain parameters are not given (VAF, IKF, VAR, IKR).

Name	Parameter	Units	Default
IS	Transport saturation current	A	1e-16
BF	Ideal maximum forward beta	-	100
NF	Forward current emission coefficient	-	1.0
VAF	Forward Early voltage	V	infinite
IKF	Corner for forward beta high current roll-off	A	infinite
ISE	B-E leakage saturation current	A	0
NE	B-E leakage emission coefficient	-	1.2
BR	Ideal maximum reverse beta	-	1
NR	Reverse current emission coefficient	-	1
VAR	Reverse Early voltage	V	infinite
IKR	Corner for reverse beta high current roll-off	A	infinite

ISC	B-C leakage saturation current	A	0
NC	B-C leakage emission coefficient	-	2
RB	Zero bias base resistance	Ohm	0
IRB	Current where base resistance falls halfway to its minimum value	A	infinite
RBM	Minimum base resistance at high currents	Ohm	RB
RE	Emitter resistance	Ohm	0
RC	Collector resistance	Ohm	0
CJE	B-E zero bias depletion capacitance	F	0
VJE	B-E built-in potential	V	0.75
MJE	B-E junction exponential factor	-	0.33
TF	Ideal forward transit time	s	0
XTF	Coefficient for bias dependence of TF	-	0
VTF	Voltage describing VBC dependence of TF	V	infinite
ITF	High current parameter for effect on TF	A	0
PTF	Excess phase at $f=1.0/(TF \cdot 2\pi)$ Hz	Deg	
CJC	B-C zero bias depletion capacitance	F	0
VJC	B-C built-in potential	Volt	0.75
MJC	B-C junction exponential factor	-	0.33
XCJC	Fraction of B-C depletion capacitance connected to internal base node.	-	1
TR	Ideal reverse transit-time	s	0
CJS	zero-bias collector-substrate capacitance	F	0
VJS	Substrate junction built-in potential	V	0.75
MJS	Substrate junction exponential factor	-	0
XTB	Forward and reverse beta temperature exponent	-	0
EG	Energy gap for temperature effect on IS	eV	1.11 (Si)
XTI	Temperature exponent for effect on IS	-	3
KF	Flicker-noise coefficient	-	0
AF	Flicker-noise exponent	-	1
FC	Coefficient for forward-bias depletion capacitance formula	-	0.5
TNOM	Parameter measurement temperature	°C	27

Temperature Effects

Temperature appears explicitly in the exponential terms.

Temperature dependence of the saturation current in the model is determined by:

$$I_S(T_1) = I_S(T_0)\left(\frac{T_1}{T_0}\right)^{XTI} \exp\left(\frac{E_g q T_1 T_0}{k\left(T_1 - T_0\right)}\right)$$

where k is Boltzmann's constant, q is the electronic charge, E_g is the energy gap, and XTI is the saturation current temperature exponent. The last two quantities are model parameters.

The temperature dependence of forward and reverse beta is given below:

$$\beta(T_1) = \beta(T_0)\left(\frac{T_1}{T_0}\right)^{XTB}$$

where XTB is a user supplied model parameter. Temperature effects on beta are carried out by appropriate adjustment to the model parameters BF, ISE, BR, and ISC.

Supported Analyses

All

General form:

```
RXXXXXXX N1 N2 VALUE
```

Examples:

```
R1 1 2 100
RB 1 2 10K
RBIAS 4 8 10K
```

`N1` and `N2` are the two element nodes. `VALUE` is the resistance in Ohm. The value can be positive or negative, but not zero.

Semiconductor Resistors

General form:

```
RXXXXXXX N1 N2 <VALUE> <MNAME> <L=LENGTH> <W=WIDTH>
+ <TEMP=T>
```

Example:

```
rload 2 10 10K
RMOD 3 7 RMODEL L=10U W=1U
```

This is a more general model for the resistor than the one presented above. This one gives you the possibility to model temperature effects and calculation of exact resistance based on geometric and process information . `VALUE` if given, defines the resistance, and geometric information will be ignored. If `MNAME` is specified, the resistance value is calculated based on information about the process in the model statement and geometric information. If `VALUE` is not given, then `MNAME` and `LENGTH` must be specified. If `WIDTH` is not given, then it will be taken form the models default width. The optional `TEMP` value is the temperature at which this device is to operate, and overrides the temperature specified in the option value.

Resistor Model

```
.MODEL {model name} R <model parameters>
```

The resistor model contains process related parameters. The resistance value is calculated from geometric information, and the value is a function of the temperature. The parameters available are shown below.

Name	Parameter	Unit	Default
TC1	First order temperature coefficient	Ohm/°C	0.0
TC2	Second order temperature coefficient	Ohm/°C^2	0.0
RSH	Sheet resistance	Ohm/Square	-
DEFW	Default width	m	1e-6
NARROW	Narrowing due to side etching	m	0.0
TNOM	Parameter measurement temperature	°C	27

The following equation is used to calculate the resistance value:

$$R = RSH \cdot \frac{L\text{-}NARROW}{W\text{-}NARROW}$$

DEFW is used to define a default value for W that is used when there is no explicit value for an element. If either RSH or L is given, a default value of 1 kOhm is used.

Temperature Effects

The effects of temperature on resistors is modelled by the formula:

$$R(T) = R(T_0)\left[1 + TC1\left(T - T_0\right) + TC2\left(T - T_0\right)^2\right]$$

where T is the circuit temperature, T_0 is the nominal temperature, and $TC1$ and $TC2$ is the first- and second order temperature coefficients respectively.

Supported Analyses

All

General form:

```
SXXXXXXX N+ N- NC+ NC- MODEL <ON> <OFF>
```

Examples:

```
s1 1 2 3 4 switch1 ON
s2 5 6 3 0 sm2 off
```

`N+` and `N-` are positive and negative nodes where the switch is connected, respectively. `NC+` and `NC-` are the positive and negative controlling nodes, respectively.

Switch Model

```
.MODEL {model name} SW <model parameters>
```

The switch model allows modelling of an almost ideal switch in AIM-Spice. The switch is not quite ideal in that the resistance can not change from 0 to infinity, but must have a finite positive value. By choosing proper values for on and off resistances, they can be effectively zero and infinite compared to other circuit elements. The available model parameters are shown below:

Name	Parameter	Units	Default
VT	Threshold voltage	V	0
VH	Hysteresis voltage	V	0
RON	On resistance	Ohm	1
ROFF	Off resistance	Ohm	1/GMIN*

An ideal switch is highly non linear. Use of switches can cause large discontinuities in node voltages. A rapidly change such as that associated with a switch changing state can cause different problems such as roundoff problems or tolerance problems that can lead to erroneous results or problems in choosing time steps. To reduce such problems follow these steps:

It is wise to set switch impedances only high and low enough to be negligible with respect to other circuit elements.

Reduce the tolerance during a transient analysis. This is done by specifying a value for TRTOL that is less than the default value 7.0. Use for example 1.0.

---

*See the section about options for a description of GMIN. The default value on GMIN results in an off-resistance of 1.0e12 ohm.

When switches are placed around capacitors, you should reduce the size of CHGTOL. Use for example 1e-16.

Supported Analyses

Distortion Analysis not supported

General form:

```
TXXXXXXX N1 N2 N3 N4 Z0=VALUE <TD=VALUE>
+ <F=FREQ <NL=NRMLEN>> <IC=V1,I1,V2,I2>
```

Example:

```
T1 1 0 2 0 Z0=50 TD=10NS
```

`N1` and `N2` are the nodes for port 1, `N3` and `N4` are the nodes for port 2. `Z0` is the characteristic impedance of the line. The length of the line can be specified in two different ways. The transmission delay, `TD` can be specified directly. Alternatively, a frequency `F` may be given together with the normalized length of the line, `NL` (Normalized with respect to the wavelength at the frequency `F`). If a frequency is specified, but `NL` omitted, 0.25 is assumed. Note that even though both ways of specifying line length is enclosed in brackets, one must be specified.

Note that this element models only one propagation mode. If all four modes are distinct in the actual circuit, then two modes may be excited. To simulate such a situation, two transmission-line elements are required.

The optional initial values consists of voltages and currents at each of the two ports. Note that these values are used only if the option UIC are specified in a transient analysis.

Supported Analyses

Distortion, Noise, and Pole-Zero Analysis not supported

General form:

```
UXXXXXXX N1 N2 N3 MNAME L=LENGTH <N=LUMPS>
```

Example:

```
U1 1 2 0 URCMOD L=50U
URC2 1 12 2 UMODL L=1MIL N=6
```

`N1` and `N2` are the two element nodes the RC line connects, while `N3` is the node which the capacitances are connected. `MNAME` is the name of the model, `LENGTH` is the length of the line in meters. `LUMPS` if given, is the number of segments to use in modelling the RC line.

URC Model

```
.MODEL {model name} URC <model parameters>
```

The model is accomplished by a subcircuit expansion of the URC line into a network of lumped RC segments with internally generated nodes. The RC segments are in a geometric progression, increasing toward the middle of the URC line, with K as a proportionality constant. The number of lumped segments used, if not specified on the URC line, is determined by the following formula:

$$N = \frac{\log\left(F_{max} \cdot \frac{R}{L} \cdot \frac{C}{L} \cdot 2\pi \cdot l^2 \cdot \frac{(K-1)^2}{K^2}\right)}{\log K}$$

The URC line will be made up strictly of resistor and capacitor segments unless the ISPERL parameter is given a non-zero value, in which case the capacitors are replaced with reverse biased diodes with a zero-bias junction capacitance equivalent to the capacitance replaced, and with a saturation current of ISPERL amps per meter of transmission line and an optional series resistance equivalent to RSPERL ohms per meter.

Name	Parameter	Units	Default
K	Propagation constant	-	2.0
FMAX	Maximum frequency	Hz	1.0G
RPERL	Resistance per unit length	Ohm/m	1000
CPERL	Capacitance per unit length	F/m	1e-15
ISPERL	Saturation current per unit length	A/m	0
RSPERL	Diode resistance per unit length	Ohm/m	0

Supported Analyses

All

General form:

```
VXXXXXXX N+ N- <<DC> DC/TRAN VALUE> <AC <ACMAG <ACPHASE>>>
+ <DISTOF1 <F1MAG <F1PHASE>>> <DISTOF2 <F2MAG <F2PHASE>>>
```

Examples:

```
vin 21 0 pulse(0 5 1ns 1ns 1ns 5us 10us)
vcc 10 0 dc 6
vmeas 12 9
```

`N+` and `N-` are the positive and negative nodes, respectively. Note that the voltage source need not to be grounded. Positive current flows from the positive node through the source to the negative node. Voltage sources can also be used as ampere meters. If you insert a voltage source with a zero value, it can be used to measure current.

`DC/TRAN` are the source value during a dc or a transient analysis. The value can be omitted if it is zero for both the dc and transient analysis. If the source is time invariant, its value can be prefixed with `DC`.

`ACMAG` is amplitude value and `ACPHASE` is the phase value of the source during an ac analysis. If `ACMAG` is omitted after the keyword `AC`, 1 is assumed. If `ACPHASE` is omitted, 0 is assumed.

`DISTOF1` and `DISTOF2` are the keywords that specify that the independent source has distortion inputs at the frequencies `F1` and `F2` respectively (see the description of the distortion analysis parameters). The keywords may be followed by an optional magnitude and phase. The default values of the magnitude and phase are 1.0 and 0.0 respectively.

All independent sources can be assigned time varying values during a transient analysis. If a source is assigned a time varying value, the value for t=0 is used during a dc analysis. There are 5 predefined functions for time varying sources: pulse, exponent, sinus, piece-wise linear, and single frequency FM. If parameters are omitted, default values shown below will be assumed. In the tables below is DT and T2 increment time and final time in a transient analysis, respectively[2].

Pulse

General form:

```
PULSE(V1 V2 TD TR TF PW PER)
```

[2] See the section describing the transient analysis

Parameters	Default values	Units
V1 (initial value)	None	Volt
V2 (pulsed value)	None	Volt
TD (delay time)	0.0	seconds
TR (rise time)	DT	seconds
TF (fall time)	DT	seconds
PW (pulse width)	T2	seconds
PER (period)	T2	seconds

Example:

```
VIN 3 0 PULSE(1 5 1S 0.1S 0.4S 0.5S 2S)
```



Sinus

General form:

```
SIN(V0 VA FREQ TD THETA)
```

Parameters	Default values	Units
V0 (offset)	None	Volt
VA (amplitude)	None	Volt
FREQ (frequency)	1/T2	Hz
TD (delay)	0.0	seconds
THETA(damping factor)	0.0	1/seconds

The shape of the waveform is:

0 < time < TD

$$V = V0$$

TD < time < T2

$$V = V0 + VA\sin\left(2\pi \cdot FREQ \cdot (time + TD)\right) \cdot \exp\left(-(time - TD) \cdot THETA\right)$$

Example:

```
VIN 3 0 SIN(2 2 5 1S 1)
```



Exponent

General form:

```
EXP(V1 V2 TD1 TAU1 TD2 TAU2)
```

Parameters	Default values	Units
V1 (initial value)	None	Volt
VA (pulsed value)	None	Volt
TD1(rise delay time)	0.0	seconds
TAU1(rise time constant)	DT	seconds
TD2 (delay fall time)	TD1+DT	seconds
TAU2 (fall time constant)	DT	seconds

The shape of the waveform is:

0 < time < TD1

$$V = V1$$

TD1 < time < TD2

$$V = V1 + (V2 - V1) \cdot \left(1 - \exp\left(-(time - TD1) \cdot TAU1\right)\right)$$

TD2 < time < T2

$$V = V1 + (V2 - V1) \cdot \left(1 - \exp\left(-(time - TD1) \cdot TAU1\right)\right) + (V1 - V2) \cdot \left(1 - \exp\left(-(time - TD2) \cdot TAU2\right)\right)$$

Example:

```
VIN 3 0 EXP(1 5 1S 0.2S 2S 0.5S)
```



Piece-wise Linear

General form:

```
PWL(T1 V1 <T2 V2 T3 V3 T4 V4 T5 V5 ....>)
```

Parameters and default values:

> Every pair of values (T_i, V_i) specifies that the value of the source is V_i at T_i. The value of the source between these values is calculated using a linear interpolation with the given values.

Example:

```
VCLOCK 7 5 PWL(0 0 1 0 1.2 4 1.6 2.0 2.0 5.0 3.0 1.0)
```



Single frequency FM

General form:

```
SFFM(V0 VA FC MDI FS)
```

Parameters	Default values	Units
V0 (offset)	None	Volt
VA (amplitude)	None	Volt
FC (carrier frequency)	1/T2	Hz
MDI (modulation index)	None	-
FS (signal frequency)	1/T2	Hz

The shape of the waveform is:

$$V = V0 + VA \cdot \sin\left((2\pi \cdot FC \cdot time) + MDI \cdot \sin(2\pi \cdot FS \cdot time) \right)$$

Example:

```
VIN 12 0 SFFM(0 1M 20K 5 1K)
```



Supported Analyses

All

General form:

```
WYYYYYYY N+ N- VNAME MODEL <ON> <OFF>
```

Examples:

```
w1 1 2 vclock switchmod1
w2 3 0 vramp sm1 ON
wreset 5 6 vclk lossyswitch OFF
```

N+ and N− are positive and negative nodes where the switch is connected, respectively. The control current is defined as the current flowing through the specified voltage source. The direction of positive control current is from the positive node through the source to the negative node.

Switch Model

```
.MODEL {model name} CSW <model parameters>
```

The switch model allows modelling of an almost ideal switch in AIM-Spice. The switch is not quite ideal in that the resistance can not change from 0 to infinity, but must have an finite positive value. By choosing proper values for on and off resistances, they can be effectively zero and infinite compared to other circuit elements. The available model parameters are shown below:

Name	Parameter	Units	Default
IT	Threshold current	A	0
IH	Hysteresis current	A	0
RON	On resistance	Ohm	1
ROFF	Off resistance	Ohm	1/GMIN*

An ideal switch is highly non linear. Use of switches can cause large discontinuities in node voltages. A rapidly change such as that associated with a switch changing state can cause different problems such as roundoff problems or tolerance problems that can lead to erroneous results or problems in choosing time steps. To reduce such problems follow these steps:

It is wise to set switch impedances only high and low enough to be negligible with respect to other circuit elements.

*See the section about options for a description of GMIN. The default value on GMIN results in an off-resistance of 1.0e12 ohm.

Reduce the tolerance during a transient analysis. This is done by specifying a value for TRTOL that is less than the default value 7.0. Use for example 1.0.

When switches are placed around capacitors, you should reduce the size of CHGTOL. Use for example 1e-16.

Supported Analyses

Distortion Analysis not supported.

General form (Level 1):

```
ZXXXXXXX ND NG NS MNAME <AREA> <OFF> <IC=VDS,VGS>
```

General form (Level 2 and 3):

```
ZXXXXXXX ND NG NS MNAME <L=VALUE> <W=VALUE> <OFF>
+ <IC=VDS,VGS> <TEMP=T>
```

Example:

```
z1 7 2 3 zm1 off
z1 0 2 0 mesmod l=1u w=20u
```

ND, NG and NS are the drain, gate and source nodes, respectively. MNAME is the model name, AREA is the area factor, and OFF indicates a optional initial value for the element in a dc analysis. If the area factor is omitted, 1.0 is assumed. The optional initial value IC=VDS,VGS is meant to be used together with UIC in a transient analysis. See the description of the .IC control statement for a better way to set transient initial conditions. The optional TEMP value for level 2 and 3 is the temperature at which this device is to operate, and overrides the temperature specified in the option value.

MESFET Model

```
.MODEL {model name} NMF <model parameters>
.MODEL {model name} PMF <model parameters>
```

In AIM-Spice there are 3 MESFET models. The difference is in the formulation of the IV-characteristics. The parameter LEVEL selects which model to use.

LEVEL=1	Statz et al.
LEVEL=2	Unified extrinsic model for uniformly doped channel (as described in Section 4.4.3)
LEVEL=3	Unified extrinsic model for delta doped channel (as described in Section 4.4.4)

Name	**Parameter**	**Units**	**Default**
VTO	Pinch-off voltage	V	-2.0 (level 1) -1.26 (level 2) -2.04 (level 3)
IS	Junction saturation current (level 1 only)	A	1e-14

BETA	Transconductance parameter (level 1 only)	A/V^2	1.0e-4
B	Doping tail extending parameter (level 1 only)	1/V	0.3
ALPHA	Saturation voltage parameter (level 1 only)	1/V	2
LAMBDA	Channel length modulation parameter (level 1 only)	1/V	0
RD	Drain ohmic resistance	Ohm	0
RS	Source ohmic resistance	Ohm	0
CGS	Zero-bias G-S junction capacitance (level 1 only)	F	0
CGD	Zero-bias G-D junction capacitance (level 1 only)	F	0
PB	Gate junction potential (level 1 only)	V	1
KF	Flicker noise coefficient (level 1 only)	-	0
AF	Flicker noise exponent (level 1 only)	-	1
FC	Coefficient for forward-bias depletion capacitance formula (level 1 only)	-	0.5

The model parameters for MESFET level 2 and 3 are the same as those for level 1, plus additional parameters listed below. Note that the following default values corresponds to n-channel MESFET devices used as example in Section 4.4 in [5].

Name	**Parameter**	**Units**	**Default**
D	Depth of device (level 2 only)	m	0.12μm
DU	Depth of uniformly doped layer (level 3 only)	m	0.035μm
JSDF	Junction saturation current density at forward bias	A/m^2	1e-2
GGR	Junction conductance at reverse bias	Ω^{-1}	40
DEL	Reverse junction conductance inverse ideality factor	-	0.04
N	Junction ideality factor	-	1
LAMBDA	Output conductance parameter	1/V	0.045 (level 2) 0.04 (level 3)
LAMBDAHF	Output conductance parameter at high frequencies	1/V	0.045 (level 2) 0.04 (level 3)
VS	Saturation velocity (level 2 only)	m/s	1.5E5

BETA	Transconductance parameter (level 3 only)	A/V^2	0.0085
VBI	Built-in voltage (level 3 only)	V	0.7
ETA	Subthreshold ideality factor	-	1.73 (level 2) 1.5 (level 3)
M	Knee shape parameter	-	2.5 (level 2) 2.2 (level 3)
MC	Knee shape parameter	-	2.5 (level 2) 2.2 (level 3)
SIGMA0	DIBL parameter	-	0.081 (level 2) 0.02 (level 3)
VSIGMAT	DIBL parameter	V	1.01 (level 2) 1.37 (level 3)
VSIGMA	DIBL parameter	V	0.1
MU	Low field mobility	m^2/Vs	0.23 (level 2) 0.2 (level 3)
ND	Substrate doping (level 2 only)	m^{-3}	3.0e23
NDU	Uniform layer doping (level 3 only)	m^{-3}	1e22
DELTA	Transition width parameter	-	5 (level 2) $\sqrt{5}$ (level 3)
TC	Transconductance compression factor	1/V	0 (level 2) 0.001 (level 3)
NDELTA	Doping of delta doped layer (level 3 only)	m^{-3}	6e24
TH	Thickness of delta doped layer (level 3 only)	m	0.01μm
ALPHAT	Temperature coefficient for VTO	V/°C	0
TBETA	Temperature coefficient for β	°C	∞
TLAMBDA	Temperature coefficient for LAMBDA	°C	∞
TETA	Temperature coefficient for ETA	°C	∞
KS	Sidegating coefficient	-	0
VSG	Sidegating voltage	V	0
TF	Characteristic temperature determined by traps	°C	TEMP
FLO	Characteristic frequency for frequency dependent output conductance	Hz	0
DELFO	Frequency range used for frequency dependent output conductance calculation	Hz	0
AG	Drain-source correction current gain	-	0

Temperature effects

Temperature appears explicitly in the exponential terms.

The dependence of the threshold voltage on temperature is modelled by the equation

$$V_T = V_{T0} - \mathrm{ALPHAT}\cdot\mathrm{TEMP}$$

The transconductance parameter BETA (for level 2 this parameter is calculated) is adjusted according to

$$\beta = \beta_0\left(1-\frac{\mathrm{TEMP}}{\mathrm{TBETA}}\right)$$

where β_0 is the model parameter BETA.

The output conductance parameter LAMBDA is adjusted according to

$$\lambda = \lambda_0\left(1-\frac{\mathrm{TEMP}}{\mathrm{TLAMBDA}}\right)$$

where λ_0 is the model parameter LAMBDA.

The subthreshold ideality factor ETA is adjusted according to

$$\eta = \eta_0\left(1-\frac{\mathrm{TEMP}}{\mathrm{TETA}}\right)$$

where η_0 is the model parameter ETA.

Supported Analyses

Level 1: All
Level 2, 3: Distortion, Noise, and Pole-Zero Analysis not supported

General form:

```
.SUBCKT SUBNAME N1 N2 N3 ...
```

Example:

```
.SUBCKT OPAMP 1 2 3 4 5
```

A subcircuit definition starts with the `.SUBCKT` statement. `SUBNAME` is the name of the subcircuit used when referencing the subcircuit. `N1`, `N2`, ... are external nodes which cannot be zero. The group of elements that follows directly after the `.SUBCKT` statement defines the topology of the subcircuit. The definition must end with the `.ENDS` statement. Control statements are not allowed in a subcircuit definition. A subcircuit definition can contain other subcircuit definition, device models, and call to other subcircuits. Note that device models and subcircuit definitions within a subcircuit definition are local to that subcircuit. They are not available outside the definition. Nodes used in a subcircuit are also local, except "0" (ground) which is always global.

.ENDS statement

General form:

```
.ENDS <SUBNAME>
```

Example:

```
.ENDS OPAMP
```

Every subcircuit definition must end with the .ENDS statement. A subcircuit name after .ENDS indicates ending of a specific subcircuit. Otherwise all definitions are ended.

General form:

```
XYYYYYYY N1 <N2 N3 ...> SUBNAME
```

Example:

```
X1 2 4 17 3 1 MULTI
```

General form:

```
.NODESET V(NODENAME)=VALUE V(NODENAME)=VALUE ...
```

Example:

```
.NODESET V(12)=4.5 V(4)=2.23
```

This control statement helps AIM-Spice locating the dc operating point. Specified node voltages are used as a first guess for the dc operating point. This statement can be useful with bistable circuits. In general .NODESET is not needed.

General form:

```
.IC V(NODENAME)=VALUE V(NODENAME)=VALUE ...
```

Example:

```
.IC V(11)=5 V(1)=2.3
```

This control statement is used to specify initial values for a transient analysis. There are two ways that this statement is interpreted depending on if UIC is specified or not.

> If UIC is specified, the node voltages in the .IC statement will be used to compute initial values for capacitors, diodes, and transistors. This is equivalent to specify IC=... for each element, but much more convenient. IC=... can still be specified and will override the .IC values. AIM-Spice will not perform any operating point analysis when this statement is used, and therefore the statement should be used with care.
>
> AIM-Spice will perform an operating point analysis before the transient analysis if UIC is not specified. The .IC statement has no effect.

Bugs Reported by Berkeley

Models defined within subcircuits are not handled correctly. Models must be defined outside of ".subckt" and ".ends" lines or the model is not found.

Convergence problems can sometimes be worked around by relaxing the maximum stepsize parameter for a transient analysis.

The base node of the bipolar transistor (BJT) is modelled incorrectly. Do not use the base node. Use a semiconductor capacitor to model base effects.

Charge storage in MOS devices based on the Meyer model is incorrectly calculated.

Transient simulation of strictly resistive circuits (typical for first runs or tests) allow a time step that is to large (e.g. a sinusoidal source driving a resistor). There is no integration error to restrict the time step. Use the maximum stepsize parameter or include reactive elements.

References

[1] B. Johnson, T. Quarles, A. R. Newton, D. O. Pederson and A. Sangiovanni-Vincentelli, *SPICE3 Version 3e1 User's Manual*, Berkeley

[2] J. S. Roychowdhury and D. O. Pederson, *Efficient Transient Simulation of Lossy Interconnect,* to appear in Proceedings of the 28th ACM/IEEE Design Automation Conference, June 17-21 1991, San Francisco.

[3] Min-Chie Jeng, *Design and Modeling of Deep-Submicrometer MOSFET's*, ERL Memo Nos. ERL M90/90, Electronics Research Laboratory, University of California, Berkely, October 1990.

[4] T. Sakurai and A. R. Newton, *A simple MOSFET Model for Circuit Analysis and its applications to CMOS gate delay analysis and series-connected MOSFET Structure*, ERL Memo No. ERL M90/19, Electronics Research Laboratory, University of California, Berkely, March 1990.

[5] K. Lee, M. Shur, T.A. Fjeldly and T. Ytterdal, *Semiconductor Device Modeling for VLSI*, 1993, Prentice Hall, New Jersey, to be published.